



## Математика для школьников 7 – 11 класса (заочный тур) Решение задачи 8. Моделирование металлических нанотрубок

1.

1.1. Радиус окружности, описанной вокруг правильного  $n$ -угольника со стороной  $a$ :

$$R = \frac{a}{2\sin(\pi/n)}.$$

1.2. В полярной системе координат, начало которой совпадает с центром трубки, координаты  $i$ -го атома:  $(R, \frac{2\pi(i-1)}{n})$ , где  $1 \leq i \leq n$ .

Тогда в декартовых координатах:

$$\begin{aligned} x &= R \cos\left(\frac{2\pi(i-1)}{n}\right) = \frac{a}{2\sin(\pi/n)} \cos\left(\frac{2\pi(i-1)}{n}\right), \\ y &= R \sin\left(\frac{2\pi(i-1)}{n}\right) = \frac{a}{2\sin(\pi/n)} \sin\left(\frac{2\pi(i-1)}{n}\right). \end{aligned}$$

2.

2.1. Шаг угловой координаты атомов (то есть, ее изменение от атома к атому) в кольцевом слое составляет  $2\pi/n$ . При размещении последующего кольцевого слоя без поворота относительно нижележащего мы не получим плотнейшего прилегания. Чтобы его достичь, последующий кольцевой слой необходимо повернуть на половину шага угловой координаты в кольцевом слое, то есть, на

$$\phi(n) = \pi/n.$$

2.2. При достаточно больших  $n$  ( $n \geq 10$ ) величина  $d$  примерно равна высоте правильного треугольника со стороной  $a$ :

$$d = \frac{\sqrt{3}}{2}a.$$


3. Алгоритм построения: по очереди строим каждую из трех трубок. Построение отдельной трубки – цикл в цикле (внутренний цикл – обход по  $j$  всех  $n$  атомов в  $i$ -том кольцевом слое и формирование координат  $(x, y, z)$ :

$$\begin{aligned} x &= \frac{a}{2\sin(\pi/n)} \cos\left(\frac{2\pi(j-1)}{n} + \frac{\pi}{n}(i-1)\right), \\ y &= \frac{a}{2\sin(\pi/n)} \sin\left(\frac{2\pi(j-1)}{n} + \frac{\pi}{n}(i-1)\right), \\ z &= \frac{\sqrt{3}}{2}a(i-1) \end{aligned}$$

внешний – последовательное увеличение номера кольцевого слоя  $i$  от 1 до  $k$ ).

Текст программы на языке [PascalABC.NET](http://enanos.nanometer.ru): см. Приложение 1.

4. В зависимости от выбранного просмотрщика и настроек размера атомов могут получаться примерно такие картинки:



## Приложение 1.

```

const
a: real = 3.32;
num: integer = 3; // число вложенных трубок

var
x, y, z: real;
sum: integer;
n: array [1..num] of integer;
k: array [1..num] of integer;
xyz: text;

begin
// задаем параметры слоев
n[1] := 24; n[2] := 17; n[3] := 10;
k[1] := 10; k[2] := 12; k[3] := 14;

// считаем суммарное число атомов
sum := 0;
for var i := 1 to num do
sum += n[i] * k[i];

// создаем файл tubes.xyz
Assign(xyz, 'tubes.xyz');
Rewrite(xyz);

Writeln(xyz, sum); // печать в файл, первая строка - значение N
Writeln(xyz); // вторая - пустая

for var l := 1 to num do // варьируем номер трубки
for var i := 1 to k[l] do // варьируем номер кольцевого слоя
for var j := 1 to n[l] do // варьируем номер атома в кольцевом слое
begin
x := Round(0.5 * a / Sin(3.14 / n[l]) * Cos(2 * 3.14 * (j - 1) / n[l] +
3.14 / n[l] * (i - 1)), 4); // 4 знака после запятой
y := Round(0.5 * a / Sin(3.14 / n[l]) * Sin(2 * 3.14 * (j - 1) / n[l] +
3.14 / n[l] * (i - 1)), 4);
z := Round(0.5 * a * Sqrt(3) * (i - 1), 4);

//печать координат в файл tubes.xyz по шаблону
Writeln(xyz, 'Au ', x, ' ', y, ' ', z);
end;

Close(xyz); // закрываем файл
end.
    
```